2BPLS-SW.sas
This routine performs a two-block partial least squares analysis (2BPLS) on two sets of variables taken from the same sample of specimens. In this example, both sets of data are Procrustes aligned shape variables, and so, following Bookstein et al. (2003), this 2BPLS can also be considered a two-block singular warps analysis. The goal of a 2BPLS is to find the linear combination of variables in each set (block) of data that maximizes covariation between the blocks of data (Rohlf and Corti, 2000). The new variables, then, tell you something about “integration” (in the broadest sense of the word) between the two blocks of data. For greater discussion of the applications, variations, and limitations of this method, see the papers already cited as well as manuscripts by Mitteroecker and Bookstein (2007, 2008) and Gunz and Harvati (2007). 
The 2BPLS proceeds by calculating a cross-covariance matrix for the two blocks of data. This is the submatrix of the standard covariance matrix that contains only those covariances between variables in one set and variables in the other set. This cross-covariance matrix is subjected to a singular value decomposition and the resulting matrices of right-singular and left-singular vectors summarize mutually predictive patterns of covariation between the two blocks of data. The first singular vectors are interpreted as a pair, each describing the variation within their respective block that covaries with that in the other block. Additional pairs of vectors are likewise interpreted, representing orthogonal axes relative to their respective within-block vectors. Unlike principal components analysis (PCA), covariation within blocks of data is not considered, and thus each matrix of singular vectors is specific to one block of data. Otherwise, this process is similar to PCA and results are typically expressed in terms of the loadings (singular vectors), the amount of variance summarized by pairs of singular vectors (eigenvalues), and specimen scores on these vectors. It is also common to calculate the correlation (r) between specimen scores on pairs of singular vectors, which gives a measure of how much the two blocks of data covary on that particular pair of singular vectors (Bastir and Rosas, 2005).  
* Caution * 
When setting up your blocks of variables, be sure to assign the set with the most variables to block1 and the set with fewer variables to block2. This requirement simplifies the programming tremendously, but if not adhered to the roles of U and V change in the singular value decomposition and the resulting singular warp scores will not be correct. There is a built-in failsafe that will check this and terminate the routine, with a warning, if the number of block1 variables is not greater than or equal to the number of block2 variables. 
INPUT: 
Unlike other routines posted on this website, this uses two data files containing numeric coordinate variables and a single character variable of specimen labels. Each row should represent the coordinates of a different specimen; specimens in both datasets should be identical and in the same order. Additional variables should be screened from the data before beginning the IML portion of this routine. Note that the use of separate data files for the two blocks of data allows one to superimpose each block of data separately by generalized Procrustes analysis (GPA). This is the standard procedure for PLS computations in other software packages (e.g., tpsPLS), and avoids the introduction of covariation between the blocks due to their relative positions within the organism. Block1.dta and block2.dta dataset have been separately superimposed for this example. Some researchers, however, may desire to use a global GPA of all landmarks before splitting them into two blocks. 
OUTPUT: 
The routine will print in the output screen the singular values, percent of covariation, and correlation for each singular warp. This table is also saved as a SAS dataset “stats.” Singular warp scores for both blocks are saved together in a SAS dataset called “sws” with column labels indicating the block of data and singular warp dimension. E.g., B1SW2 is the second singular warp from block 1. Singular warp scores can be plotted in Solutions\Interactive Data Analysis. Finally, the matrices of singular vectors are saved in SAS datasets B1vectors and B2vectors. 

Potential sources of error:

· The number of variables in block1 must be equal to or greater than the number in block2. See CAUTION above.

· The exact same specimens must be in both blocks of data, and they must be listed in the same order. This latter problem can be fixed using PROC SORT.  

· Users should determine whether their research question is better addressed by superimposing all of the landmarks and then dividing them into blocks or by separately superimposing the blocks of data. This example is set up to input two blocks of data superimposed separately. This is not an arbitrary decision and should be carefully considered. 
· Before running the IML portion of this routine, both datasets should be screened down to only those numeric variables that are to be used in the 2BPLS. ANY numeric variables, including numeric taxon or sex indicators, will be indiscriminately read into IML and incorporated into the analysis if they are in the block1 or block2 data sets. 

Code annotation: 
### indicates lines that need to be changed for different datasets
data one; infile 'C:\...\block1.nts'
### names dataset “one” and designate the path to the file from which the new data will be read
firstobs=2 obs=41; 
### “firstobs=” indicates that the first datum should be read from the second line of the file, and “obs=” tells SAS to continue reading until the 41st line

input cat $; 
the “input” statement initiates the data reading and this is followed by the list of all variables to be read. In this case, the variable “cat” is the only variable inputted, and the “$” tells SAS that “cat” is a character variable (rather than a numeric). 
run;
executes the datastep
data block1; set one; 
names dataset “block1” using all of the data from dataset “one”
infile 'C:\...\block1.nts'
### designates the path to the file from which the new data will be read. Under most circumstances, this should be identical to the path in data step “one.”
firstobs=43; 
### “firstobs=” indicates that this data step will begin reading data on the 43rd line of the file. Because we want to read in all of the remaining data in the file, no “obs=” statement is needed. 
input x1-x30; 
### this “input” statement indicates that 30 variables will be read in and named x1, x2, x3, …x30. SAS will therefore assign the first thirty variables to the first specimen, the next thirty to the second specimen, etc. 

run;
executes the datastep

data two; 
### These data steps follow
infile 'C:\...\block2.nts'
### the same protocol
firstobs=2 obs=41; input cat $; run;
### as those above and
data block2; set two; 
### can be modified
infile 'C:\...\block2.nts'
### appropriately using
firstobs=43; input x1-x18; run;
### those instructions.
proc iml; 
begins processing commands in Interactive Matrix Language
DF=7;
### This number is used to restrict the number of singular warp statistics given according to the degrees of freedom lost during superimposition. It should be “7” for 3D data and “4” for 2D data. This is largely arbitrary, however, as one should not be trying to interpret the high-rank singular warps.
use block1; read all into X1; 
“use block1” designates the data set work.block1 as the source for new data. “read all in X1” causes all of the NUMERIC variables in block1 to be put into a new matrix “X1”.

use block2; read all into X2;
“use block2” designates the data set work.block2 as the source for new data. “read all in X2” causes all of the NUMERIC variables in block2 to be put into a new matrix “X2”.
N=nrow(X1); 
“N” is defined here as the total number of specimen, calculated as the number of rows in the matrix “X1”

LM1=ncol(X1); 
“LM1” is defined here as the total number of variables in block1, calculated as the number of columns in the matrix “X1”

LM2=ncol(X2);
“LM2” is defined here as the total number of variables in block2, calculated as the number of columns in the matrix “X2”
if LM1<LM2 then do; 
This DO-loop is the failsafe so that this routine is terminated if the number of variables in block1 is less than the number in block2. In this case, LM1 will be less than LM2 and the “if” will be satisfied, causing the statements in the DO-loop (i.e., between “do;” and “end;”) to be executed.
print '*****  ...   *****',
This is the warning message printed when the
      '*****  ...   *****',,
routine terminates. Commas “,” tell SAS to
      'Switch ...re-run.'; 
continue printing on the next line.
end;
This ends the DO-loop
else do;
“else” here provides an alternative direction if the above “if” statement is not satisfied. I.e., if LM1 is NOT less than LM2, the new DO-loop defined here will activate, running the rest of the PLS analysis. 
lab1='B1SW1':'B1SW18'; 
### This statement sets up the labels for the block1 singular warps scores. The highest number (in this example “18”) should be changed to reflect the number of variables in BLOCK2.
lab2='B2SW1':'B2SW18';
### This statement sets up the labels for the block2 singular warps scores. The highest number (in this example “18”) should also here be changed to reflect the number of variables in BLOCK2.
labSW=lab1||lab2;
The double pipe “||” merges horizontally the two matrices of labels creating a new matrix, “labSW”, with all of the labels.
meanX1=X1[+,]/N; 
“meanX1” is a row vector of means for each coordinate, computed by summing the columns of “X1” and then dividing the resulting columns by the number of specimens (“N”).
X1bar=shape(meanX1,N,LM1); 
“X1bar” is a matrix with rank equal to “X1” (e.g., N x LM1) with the coordinate means (“meanX”) repeated in every row.
Dif1=X1-X1bar;
“Dif1” is the mean-centered (by variable) matrix for block1 data. It is created here by subtracting the matrix of mean values (X1bar) from the block1 matrix (“X1”).
meanX2=X2[+,]/N; 
This series of commands creates the mean-
X2bar=shape(meanX2,N,LM2); 
centered matrix for block2 (“Dif2”)
Dif2=X2-X2bar;
in the manner described above.
CCV=(Dif1`*Dif2)/(N-1);
CCV is the cross-covariance matrix containing only those covariances between a variable in block1 and a variable in block2.
call SVD(u,q,v,CCV);
“call SVD” initiates a singular value decomposition of the last matrix listed in the parentheses (“CCV”). Columns of “u” will be the singular vectors for block1. Columns of “v” will be the singular vectors for block2, and “q” is a matrix of singular values that correspond to the matching vectors in both “u” and “v”. These singular values are the square roots of the eigenvalues for those vectors. 
q=q[1:LM2-DF]; 
This re-defines “q” (the singular values) to only include those values corresponding to the degrees of freedom in the smallest block.
q2=q##2; 
“q2” is defined here as the square of the values in “q”, and therefore is a matrix of the eigenvalues.
var=q2[+,]; 
The only column in q2 is added here representing the sum of all eigenvalues, “var”.
pcv=j(nrow(q),1); SW=j(nrow(q),1);
These statements define two new matrices (“pcv” and “SW”) with as many rows as “q” and only a single column. The “j” function used here fills these matrices with “1”s.
do i=1 to nrow(q); 
This initiates a new DO-loop that will keep iterating equal to the number of rows in “q”. For each iteration, the value of “i” will change, starting at 1 and adding 1 more each time it repeats. 
SW[i,]=i; 
Each iteration of this loop will go to the ith row of SW and change the value there to whatever is the current value of “i”.
pcv[i,1]=q2[i,]*100/var; 
“pcv” is here populated with the eigenvalues listed as a percent of the total covariance (sum of the eigenvalues). This is done by taking every value in “q2”, multiplying it by 100 and dividing it by “var”. 
end;
terminates the DO-loop
swsX1 = Dif1*u; swsX2 = Dif2*v;
Singular warps scores are computed here by multiplying the mean-centered block1 (“Dif1”) and block2 (“Dif2”) data matrices by the block1 and block2 eigenvectors (“u” and “v” respectively). Singular warps scores for block1 are stored in “swsX1” and singular warps scores for block2 are stored in “swsX2”.
sws=swsX1||swsX2; 
This merges horizontally the “swsX1” and “swsX2” singular warps scores matrices.

r=corr(sws);
The correlation module (“corr”) used here computes the matrix correlation for the singular warps scores (“sws”) and stores the correlations in “r”. 
r=r[1:LM2-DF,LM2+1:2#LM2-DF]; 
Since our only interest is the correlation between pairs of singular warps (e.g., the first singular warps of block1 and block2), this statement cuts out the parts of the r matrix that are extraneous, leaving the correlations in the diagonal of “r”.
r=vecdiag(r); 
This function extracts the diagonal of “r” and puts it into a column vector, also labeled “r”.
labstat={'SW' ... 'corr. (r)'};
“labstat” is a matrix of labels for the columns of the “stat” matrix (see below)
stat=SW||q||pcv||r;
“stat” is an amalgamation to the “SW” (numbers of the singular warps), “q” (singular values), “pcv” (percent of the total covariance), and “r” (correlations between paired singular warps) matrices. The double pipe (“||”) horizontally merges these together.
reset autoname;
the “reset” function has many uses, and here the “autoname” option is used to space apart the columns of “stat” in the output window
print stat[colname=labstat];
This prints to “stat” matrix in the output window with the “labstat” matrix providing labels for the columns.
create stats from stat[colname=labstat]; 
“create” generates a regular SAS dataset outside of the IML arena. Here it is naming that dataset “stats” and using the IML matrix “stat” to create it with column labels (in this case variable names) given in the “labstat” matrix.

append from stat;
“append” actually transfers to data from “stat” to the open dataset (“stats”)
create sws from sws[colname=labSW]; 
These commands create datasets, in the manner
append from sws; 
described above, for the the singular warps
create B1vectors from u; append from u;
scores (“sws”), and the eigenvectors of both
create B2vectors from v; append from v;
block1 and block2 (B1vectors, B2vectors).
end;
This ends the “ELSE DO” loop initiated above.
abort;
“abort” terminates the IML processing. 
data singwarps; merge one sws; 
“singwarps” is a new dataset that merges the labels from dataset “one” (note, these should be identical to those in dataset two) with the singular warps scores in “sws”.
sps=substr(cat,1,2); 
### For graphing purposes, the variable “sps” is created here from the first and second characters in the “cat” variable. 

sex=substr(cat,4,1); 
### For graphing purposes, the variable “sex” is created here from the fourth character in the “cat” variable.
run;
This executes the data step.
